
Exercise 2 
 

Basic Multinomial Choice 
 

=========================================================================== 

This assignment will consist of some exercises with the multinomial logit models.  The data set for the 

exercises is the multinomial choice file 

 

    mnc.lpj 

 

This project file contains both the choice data sets discussed in class, the brand choices data and the travel 

mode data.  Altogether, there are 12,800 observations in the brand choices data.  The travel mode data 

appear in the first 840 rows of the data area.   

=========================================================================== 

 

These computations will be based on the shoe brand data. 

 

Use  SAMPLE ; 1 – 12800 $ 

 
Preliminaries:  The exercises will build on the basic model 

 
 U(brand)   =  1Fashion + 2Quality + 3Price + 4Price

2
  + brand, 

 U(none) =   + none. 

 
The model analyzes 4 choices, brand1, brand2, brand3, and ‘none of the brands.’  ASC4 is the constant 

term that appears only in the ‘NONE’ choice.  The central model command will be 

 

 NLOGIT  ; Lhs = choice  

; Choices = brand1,brand2,brand3,none 

   ; Rhs = fash, qual, price, pricesq,asc4 $ 

 

There are two characteristic variables, gender and age.  Gender is coded as MALE = 1 for men, 0 for 

women.  AGE is categorized as AGE25 = 1 if age < 25 and 0 else, AGE39 if 25 < age < 39 and 0 else, 

and AGE40 if age > 40. For modeling purposes, we will drop AGE40. 

 
I.  Multinomial Choice 
 

1.  Fit the basic model.  Is pricesq significant?  Use the Wald test based on estimates of the basic model.  

Fit the model without pricesq and use a likelihood ratio test. 

 

2.  Do age and sex matter?  Add age24, age39 and male to the basic model and use a likelihood ratio test.  

Note that these variables are choice invariant, so they must be added as 

 

  ; Rh2 = male, age24, age39 

 

If you add them to the Rhs list instead, estimation will break down. 

 

3.  Is there a more general age difference in utility?  To explore this, use a Chow style test.  To start, 

 

CREATE ; Young = age25 $ 



Then, 

NLOGIT ; if [young = 1] … $ 

 CALC ; lyoung = logl $ 

 NLOGIT ; if [young = 0] … $ 

 CALC ; lold = logl $ 

 NLOGIT ; … $ 

 CALC ; list ; lrtest = 2*(lyoung+lold – logl) $   

 

Is the statistic larger than the critical value?  Note, there is a way to combine all of these in a single 

operation: 

 

 NLOGIT ; For[(test) young =*,0,1]  

; … your model specification (must not include age25) 

$ 

 

Carry out the same structural form test for men (MALE=1) vs. women (MALE=0). 

 

II.  Elasticities and Marginal Analysis 
 

We estimate a marginal effect (of price) in the MNL model.  What are the estimates of the own and 

cross elasticities across the three brands?  What is the evidence of the IIA assumption in these results?  

Try the following: 

 

NLOGIT  ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

  ; Rhs = Fash,Qual,Price,Asc4  

  ; Effects : Price (*) $ 

 

What are the effects?  Note that the squared price is not in the equation.  It is unclear how to compute the 

elasticity directly in the presence of the square.  But, there is a way to explore the effects with the 

simulator.  Consider the following: 

 

NLOGIT  ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

  ; Rhs = Fash,Qual,Price,Pricesq,Asc4 $ 

NLOGIT  ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

  ; Rhs = Fash,Qual,Price,Pricesq,Asc4  

; Simulate = * ; Scenario:  

price (brand*) = [*] 1.1/ 

pricesq(brand*)=[*]1.21 $ 

 

The scenario increases the price by 10% and consequently, the square by 21%.  What happens to the 

market shares under this scenario?  Try a larger price increase, say 25% (and 62.5%). 

 

  



Impact of a price change.  What would happen to the market shares of the three brands if the price of 

Brand 1 of shoes rose by 50%.  What would happen to the market shares if the prices of all three brands 

rose by 50% 

 

CLOGIT ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

       ; Rhs = Fash,Qual,Price,Asc4  

       ; Effects : Price (*) $ 

CLOGIT ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

       ; Rhs = Fash,Qual,Price,Asc4  

       ; Simulation = *  

       ; Scenario: Price (Brand1)  = [*] 1.5 $  

CLOGIT ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

       ; Rhs = Fash,Qual,Price,Asc4  

       ; Simulation = *  

       ; Scenario: Price (Brand1,Brand2,Brand3)  = [*] 1.5  $ 

 

III.  Specification Analysis 
 

A.  Testing for IIA.  Is Brand3 an irrelevant alternative in the choice model?  Given the way the data are 

constructed, one wouldn’t think so.  Here we investigate.  Carry out the Hausman to test the IIA 

assumption using Brand 3 as the omitted alternative.  What do you find? 

 

? 

? Testing for IIA 

? 

SAMPLE ; All $ 

CLOGIT ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

       ; Rhs = Fash,Qual,Price,Asc4  $ 

CLOGIT ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

       ; Rhs = Fash,Qual,Price,Asc4 ; IAS = Brand3  $ 

 

B.  Functional form and marginal impact.  Do men pay more attention to fashion than women?  To 

investigate, we fit the choice model with a different coefficient on fashion for men and women.  Then, 

simulate the model so as to see what happens when the variable which carries this effect into the model is 

zero’d out.  What are the results?  How do  you interpret your findings? 

 

? 

?  Do men pay different attention to fashion than women? 

?  Is the difference statistically significant? 

? 

Create ; MaleFash = Male*Fash $ 

CLOGIT ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

       ; Rhs = Fash,Qual,Price,Asc4,MaleFash $ 

CLOGIT ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None 

       ; Rhs = Fash,Qual,Price,Asc4,MaleFash  

       ; Simulation = * ; Scenario: MaleFash(*) = [*] 0 $ 

 

  



C.  Multinomial Probit Model.  Do the multinomial logit and multinomial probit models give similar 

results?  You can’t tell directly from the coefficient estimates because of scaling and normalization, so 

you have to rely on other indicators such as marginal effects.  Fit a multinomial probit and a multinomial 

logit model, and compare the results.  Note, estimation of the MNP model is extremely slow, so we have 

set it up with a very small number of replications and stopped the iterations at 5.  This particular model 

would take 30-50 iterations, and an hour or two, to finish.  As it is, this will take a few minutes. 

 

NLOGIT   ; Lhs = Mode ; output=ic 

  ; Choices = Air,Train,Bus,Car 

  ; Rhs = TTME,INVC,INVT,GC; Rh2=One,Hinc  

  ; Effects: invc (*) $ 

 

NLOGIT   ; Lhs = Mode ; MNP ; PTS = 5 ; Maxit = 5 ; Halton 

  ; Choices = Air,Train,Bus,Car 

  ; Rhs = TTME,INVC,INVT,GC; Rh2=One,Hinc  

  ; Effects: invc (*) $ 

 
The multinomial probit model is distinguished by allowing the correlations across utility functions and to 

some degree, heteroscedasticity across utilities – though not across individuals.  For a four outcome 

model, the covariance matrix for a multinomial probit model must be of the form 

  

11

21 22

31 32 1

0 0 0 1

 
 
 
  
  
 
 

 

 

The restrictions on the last two rows are normalizations needed for identification.  The built in 

multinomial probit estimator in Stata actually goes further, and imposes  = I.  You can replicate this 

model by adding 

 

  ;  SDV = 1 ; COR = 0  

 

To the multimomial probit command.  Fit this model, then compare the reported elasticity matrix to that 

reported by the initial multinomial logit model.  (The difference mostly reflects scale differences in the 

coefficients – this is the familiar relationship between probit and logit models.) 

 

IV.  Nested Logit Model 
 

A. Nested logit model.  We begin with a simple nested logit model. 

 

? 

? Basic nested logit 

? 

NLOGIT  ; Lhs = Mode ; Choices=Air,Train,Bus,Car 

        ; Rhs = One,GC,TTME,INVT,INVC 

        ; Tree = Private(Air,Car),Public(Train,Bus) 

        ; Show Tree $ 

CALC    ; LOGLU = LOGL $ 

 



B.  Constrained nested logit model.  Constraining the IV parameters to equal 1 returns the original 

multinomial logit model.  Use this device to test the restriction.  Note that this specification test is 

whether the MNL is appropriate, against the alternative of the nested logit model. 

 

? 

? Constrain IV parameters to produce MNL model 

? 

NLOGIT  ; Lhs = Mode ; Choices=Air,Train,Bus,Car 

        ; Rhs = One,GC,TTME,INVT,INVC 

        ; Tree = Private(Air,Car),Public(Train,Bus) 

        ; IVSET:(Private,Public)=[1] $ 

CALC    ; LOGLR = LOGL $ 

CALC    ; List ; LRTEST = 2*(LOGLU - LOGLR) $ 

 

C.  Degenerate branch.  A branch that contains only one alternative is labeled ‘degenerate’ (for reasons 

lost to antiquity).  The RU1 and RU2 normalizations produce different results for such models.  Fit the 

two and examine the effect. 

 

? 

?  Degenerate branch.  Two normalizations 

? 

NLOGIT  ; Lhs = Mode ; Choices=Air,Train,Bus,Car 

        ; Rhs = One,GC,TTME,INVT,INVC 

        ; Tree = Fly(Air),Ground(Car,Train,Bus) $ 

NLOGIT  ; Lhs = Mode ; Choices=Air,Train,Bus,Car 

        ; Rhs = One,GC,TTME,INVT,INVC 

        ; Tree = Fly(Air),Ground(Car,Train,Bus) ; RU2 $ 

 

D.  Alternative approaches to reveal scaling.  The nested logit model can be modified to act like the 

heteroscedastic extreme value buy making all branches contain one alternative.  This will allow a 

different scale parameter in each branch.  The HEV model is another way to do this.  Are the results 

similar? 
 

? 

? Use nested logit to reveal scaling. 

? 

NLOGIT  ; Lhs = Mode ; Choices=Air,Train,Bus,Car 

        ; Rhs = One,GC,TTME,INVT,INVC 

        ; Tree = Fly(Air),Drive(Car),Rail(Train),Ride(Bus)  

        ; IVSET: (Ride) = [1] ; Par $ 

NLOGIT  ; Lhs = Mode ; Choices=Air,Train,Bus,Car 

        ; Rhs = One,GC,TTME,INVT,INVC 

        ; HET ; SDV = SA,ST,1.0,SC $ 
 


